

~~SEC FC~~
USGS
OFR
79-1434

GL00084

U.S. Department of the Interior
U.S. Geological Survey
Denver, Colorado 80225

A GENERALIZED PROGRAM IN HEWLETT-PACKARD BASIC FOR
CREATING, EDITING, AND PRINTING OUT DATA FILES FOR GEOCHEMICAL APPLICATIONS

by

Kenneth R. Ludwig

Open-File Report 79-1434
1979

UNIVERSITY OF UTAH
RESEARCH INSTITUTE
EARTH SCIENCE LAB.

A GENERALIZED PROGRAM IN HEWLETT-PACKARD BASIC FOR
CREATING, EDITING, AND PRINTING OUT DATA FILES FOR GEOCHEMICAL APPLICATIONS

by

Kenneth R. Ludwig

INTRODUCTION

The following mini-computer program is designed to provide a flexible, easy-to-use program for the purpose of creating and using relatively small (1,750 elements) data files. These data files can then be used to produce hard-copy tables using this program, camera-ready plots using other plotting programs, or for a variety of statistical manipulations. The program is written in Hewlett-Packard BASIC for a Hewlett-Packard Model 9831 desktop computer with 24K bytes of memory, a model 9866 thermal printer and an (optional) model 9871 impact printer. A model 9830 computer with the appropriate string-variable ROM and memory, or models 9835-9845 computers could also be used with slight modifications to the program.

The data are stored in a 50 x 35 matrix with split (6-digits, 10^{-63} - 10^{+63} range) precision accuracy. Data for up to 50 samples, with up to 35 parameters per sample, may be stored in each data file. Each data file is assigned a name of up to 80 characters, each sample a name of up to 36 characters, and each parameter a name of up to 24 characters. Once a data file has been created, the data may be edited and re-ordered in a variety of ways via a few simple commands. The printout may be done rapidly with the thermal printer; or, using the impact printer, more slowly but with user-defined number of significant figures, with uncertainties printed out to the right of some or all of the data-elements, and with numeric sub- or super-scripts for the parameter names. Errors resulting from improper or illogical input are, in most cases, automatically

identified and explained, without interruption of the program operation. Data files are stored in 5000-word files on standard Hewlett-Packard data cartridges, so that each cartridge can hold over 30 data files totalling over 50,000 entries.

Instructions for use of the program (appendices I and II) are stored as a separate program requiring a 5000-word file on the program cassette. Because this instruction program can be accessed by the main program, the file number for this program must be contained within the main program. As listed in the appendices, the main program is in file 1, track 0, and the instruction program in file 2, track 0. These may be changed by changing the values specified in line 440 of the main program (appendix III), and lines 30 and 1740 of the instruction program (appendix II).

I will be happy to copy the programs onto an appropriate data-cartridge, supplied by any interested person for either HP-9830 or HP-9831 computers.

FIGURE CAPTIONS

Figure 1. Sample printout using option 4 and HP-9866b printer. The printed line defining option 15, and the two lines defining the location, number of defined sets and parameters, and the data-file name always appear. All significant figures (up to 6) originally entered into the data file are printed out.

Figure 2. Sample printout using option 11 and HP-9871 impact printer. Set and parameter numbers were suppressed, and numeric superscripts appear in the parameter names. Significant figures were restricted by the user to three or four, depending on the parameters, and printout of uncertainties was requested. Character spacing is slightly reduced and displaced to the left for printout of these parameters to avoid interference with the printout of adjacent numbers.

TABLE 1 - SAMPLE PRINTOUT USING OPTION 4 AND HP-3866B THERMAL PRINTER

DATA FROM TRACK 1 FILE 3 (47 SETS, 34 PARAMETERS DEFINED)
 MIDNITE MINE ORES & MINERALS, AUTOREDUCED 9/10/79

(parameter #)	1	2	15	16
SET#	SAMPLE NAME SAMPLE WT. (GRAMS)	U (%)	$^{235}\text{U} \times 204\text{Pb}$	MU-NU % ERROR
8	169	1.104	6.837011	6.851186
9	284-256	1.76	6.390719	6.875265
10	146B	0.8231	4.46433	0.837744
11	222 #1	1.0363	12.6731	3692.53
12	443-199.5	1.4234	12.3821	1.03135
13	351-405	1.1487	3.27042	544.996
14	382-292	1.5182	17.2686	680.774
32	146B P/C #10 9.000E-03 residue.	7.055E-04	0.294358	0.833384
38	382-291 PY#2 1.900E-03 (in P/C)	0.314456	3.6684	0.74346
26	146B P/C #11 6.560E-05	58.9752	571.194	1.68043
27	146B P/C #13 9.560E-05	56.1993	3680.12	3.91161
28	146B P/C #14 9.160E-05	63.2595	463.638	1.54774
45	MIDNITE MINE 2.600E-03 KP ZIRCONs mag #1	0.454337	5768.08	61.0748
2	336 #1	0.7602	6.0224164	0.750451

TABLE 2 - SAMPLE PRINTOUT USING OPTION 11 WITH HP-9871 IMPACT PRINTER

SAMPLE NAME	SAMPLE WT. (GRAMS)	Th (PPM)	206Pb/204Pb BLANK-CORR.	207Pb/204Pb BLANK-CORR.	207Pb/235U RAD.	207Pb/235U AGE (M.Y.)	235U/204Pb	U (%)
10S	1.1	---	109.5 ± 0.3	21.50 ± 0.06	0.0400	39.8 ± 2.3	144 ± 1	0.837
284-256	1.8	---	276.6 ± 0.8	22.01 ± 0.06	0.0270	27.1 ± 1.4	232 ± 2	0.391
146B	0.82	15.1	450.8 ± 1.2	38.22 ± 0.11	0.0504	49.9 ± 0.8	447 ± 4	4.46
222 #1	1.0	1.60	3104 ± 12	174.9 ± 0.7	0.0515	50.9 ± 0.4	3090 ± 26	12.7
443-199.5	1.4	1.37	1454 ± 4	93.50 ± 0.28	0.0516	51.1 ± 0.6	1510 ± 16	12.4
MIDNITE MINE KP ZIRCONS nonmag #1	0.0045	---	5937 ± 2179	302.1 ± 105.5	0.0662	65.1 ± 0.4	4330 ± 1594	0.310
MIDNITE MINE KP ZIFCONS nonmag #2	0.0033	795	4239 ± 794	220.1 ± 38.5	0.0797	77.9 ± 0.7	2560 ± 483	0.425
MIDNITE MINE KP ZIRCONS mag #1	0.0026	884	9364 ± 5707	467.5 ± 276.1	0.0783	76.6 ± 0.6	5770 ± 3523	0.454
MIDNITE MINE MONAZITE	0.0075	77800	341.8 ± 11.8	31.12 ± 0.58	0.0763	74.6 ± 3.9	202 ± 13	0.307
TURTLE LAKE KP ZIRCONS	0.0050	---	3389 ± 426	180 ± 21	0.0960	93.1 ± 0.6	1710 ± 217	0.229
74	0.76	---	21.12 ± 0.06	15.80 ± 0.05	0.123	118 ± 509	0.609 ± 0.005	5.58E-04
336 #1	0.76	---	84.99 ± 0.21	15.79 ± 0.04	0.00706	7.15 ± 39.13	8.34 ± 0.06	0.0224
72 #1	0.85	---	67.43 ± 0.23	15.59 ± 0.04	-0.0258	-26.6 ± 63.2	5.25 ± 0.04	0.00707
530	0.65	8.74	18.96 ± 0.04	15.62 ± 0.04	---	---	0.133 ± 0.001	4.26E-04

APPENDIX I -- INSTRUCTIONS PRINTOUT FROM INSTRUCTIONS PROGRAM

INSTRUCTIONS FOR KRL 50X35 DATA-FILE PROGRAM

INTRODUCTION

This program permits the user to use a 50X35 matrix to store split-precision (6-digit accuracy) data for up to 50 samples ('sets') and up to 35 variables ('parameters') per sample, with user-defined sample and parameter names. Once stored, the data may be edited, printed out, added to, manipulated, and plotted via a number of subroutines and plotting programs. If an HP-9871 impact printer is available, the data may also be printed out in hard-copy table form, including parameter names with superscript or subscript (i.e., isotope/oxide form) notation, with user-defined number of significant figures, and with associated errors printed next to the proper values.

GENERAL COMMENTS

If you intend to store a data-file, make sure a premarked, empty or expendable file of at least 5000 words length is available.

Most displayed queries may be answered with a space in order to escape from that subroutine.

If an HP-9871 impact printer is connected to the computer and is turned on, the data-printout will automatically be done on the impact printer. The (standard) model 9866 thermal printer must have a select code of 2, the impact printer a select code of 6 (assigned in lines 50 and 60 of the program).

CREATING A DATA-FILE (OPTION 1)

DISPLAY: FILE NAME?

Input a name (<= 80 characters long) to be assigned to the file.

DISPLAY: PARAMETER 1 NAME?

The parameter names may be up to 2 lines of 12 characters, so this query requires two responses. A question-mark will prompt input of the second line. If you wish the parameter name to be printed out with all numerals as superscripts, input ^ as the first character, first line (e.g., ^87Rb/86Sr). If numerals are to be printed out as subscripts, input * as the first character of the line (e.g., *Al2O3). Superscripts and subscripts will only appear on impact-printer printouts and plotter-drawn plots. After each

parameter name has been entered, the display will request the name of the next parameter. When no more parameters are to be named, answer the two parameter-name queries with a space.

DISPLAY: SET # 1 NAME?

The set (sample) names are stored as three lines of twelve characters each, so that three display queries must be answered. When no more sets are to be named, simply answer the first query with a space. Set names are requested before the data for each set is to be entered.

DISPLAY: SET 1 , % A1203?

The parameter name here is of course an example. Input a value for the particular set and parameter requested. The next query will be the next-numbered parameter for that set, and so on until the query will ask for data for the next-numbered set. To cease data-input, enter a space in response to the first request for a set name. If no value for this data-element is known or relevant, enter the number 1E50. The printout for this number will then be three dashes (---).

DISPLAY: STORE IN TRACK#, FILE #?

If you do not wish to store the data yet, input a space. Otherwise, enter the track # and file # of the already-marked, >5000-word file. Make sure the cassette tab is at the RECORD position first, though.

TO CREATE A DATA-FILE USING PARAMETERS DEFINED IN AN EXISTING DATA-FILE (OPTION 0)

This option permits the use of parameters whose name and numbers have already been defined in the creation of a previous data-file. Operation is straightforward.

PRINTOUT OF THE DATA-FILE (OPTIONS 4 AND 11)

Option 4 is a 'scratchpad' printout which prints out the data as input, with numbered parameters and sets. Option 11 permits a formal, hard-copy table to be produced.

OPTION 4:

DISPLAY: DATA FROM TRACK#, FILE #?

If the data-file is already in memory, answer with a space. If not, make sure that the tape cartridge of interest is in the computer. Input the appropriate track and file numbers.

DISPLAY: PARAMETER #?

You may print up to 4 parameters on a single sheet of paper using the 9866 thermal printer, or up to 8 parameters using the 9871 impact printer and 15-inch wide paper. Enter the parameter numbers, separated by commas, that you wish printed out.

If the data is being printed out on the impact printer, with the number of significant figures defined by the user (option 11), the uncertainties of each data-element can be printed out just to the right of the data-element itself. The uncertainties must be stored in the data-file as a specified parameter. In response to the 'PARAMETER #?' query, enter a slash (/) after each appropriate parameter, followed by the parameter number of the associated uncertainties. For example, an input of 2/17,6,3/19 would result in a printout of the uncertainties to parameters 2 and 3 (stored as parameters 17 and 19) being printed out, with a +/- symbol, to the right of the appropriate data-elements, and with the same number of decimal places as the data-element. If the uncertainties are stored as percent, they will automatically be converted to absolute values, so long as '%', 'PERCENT', or 'percent' appears in the parameter name. Note that if too many characters are required for the data-printout in this manner, interference with the next column of data may occur. If the uncertainty is assigned a value of zero, no uncertainty printout will occur.

DISPLAY: # OF SIGNIFICANT FIGURES?

This display appears if you chose printout option 11 instead of 4. Enter the maximum number of significant figures (from 1 to 6) that you wish printed out for each parameter chosen, in the order that you entered the parameter numbers. If all parameters are to have the same number of significant figures, you need only enter that number once. The data will be rounded off in the printout, with least significant digits of 5 rounded to the nearest even digit.

DISPLAY: SET(S) TO BE PRINTED OUT?

If sets are to be printed out in serial order, input first last set numbers, separated by a comma. If sets are to be printed out in arbitrary order, input the set numbers in the desired order of printout, separated by semicolons. The two modes may be combined in one response - e.g., an input of 2,5;24;11,9;16;1 would

result in printout of sets in the order 2,3,4,5,24,11,10,9,16,1. Note that sets may be printed out in descending as well as ascending serial order. If 'ALL' or 'all' is the response to the query, all of the defined sets will be printed out in order. Data from different data-files (with the same parameters and parameter numbers) can be printed out in the same table by placing an asterisk followed by the track and file numbers of the new data-file, before the set numbers of interest. Thus if sets 7 through 12 from the data-file in track 1, file 23 were to be added to the printout following sets 43 and 12 of a data-file in memory, the response to the 'SET(S)' query would be 43;12;*1,23,7,12.

OPTION 11:

DISPLAY: PARAM. HEADINGS- 1=YES, 2=NO?

An input of 1 will give a table with parameter and set names, but not parameter or set numbers, printed out. An input of 2 will suppress the printout of parameter names, in case data from more than one data-file is to be printed out, or if vertical juxtaposition of different sheets of printout is desired.

TO EDIT DATA (OPTION 3)

DISPLAY: EDIT SET #S?

Enter the set numbers, separated by commas, of the sets to be edited. Thus if you wanted to change some of the values of sets 2, 41, and 16, the input would be 2,41,16.

DISPLAY: EDIT PARAMETER #S?

Enter the parameter numbers whose values, for the sets chosen above, to be changed, separated by commas. To edit a sample name, 0 must be the first number entered. To change all of the parameter values, enter ALL (or all). Thus to edit the sample name and all of the parameter values, the input would be 0,ALL. Enter a space to escape.

TO DELETE DATA SETS (OPTION 5)

DISPLAY: SET(S) TO BE DELETED?

If only one set is to be deleted, input that set number.

If a block of sets in sequence are to be deleted, input the first and last numbers of the sequence, with a comma separating the two. The sets are renumbered after each deletion. Input a space after the desired deletions have been accomplished.

TO INSERT A DATA-SET (OPTION 6)

DISPLAY: INSERT DATA BEFORE SET #?

Input the number of the set before the new set is to be placed. The usual data-input routine then occurs, and sets higher-numbered than the inserted set are renumbered.

TO ADD A PARAMETER NAME (OPTION 9)

This option permits you to increase the number of defined parameters by defining additional parameter names (up to 35 total). After adding a parameter name, all the values of the new parameter will be zero. To input the corresponding data, you will need to:

CHANGE ALL VALUES OF ONE PARAMETER (OPTION 10)

This option permits you to change the values of a given parameter for all of the sets -- i.e., to enter data by column instead of by row.

TO PRINT OUT A LIST OF PARAMETER NAMES (OPTION 12)

TO PRINT OUT A LIST OF SET NAMES (OPTION 13)

TO PRINT OUT A LIST OF OPTIONS (OPTION 15)

These operations are straightforward.

APPENDIX II -- LIST OF INSTRUCTIONS PROGRAM

```
10 REM INSTRUCTIONS FOR KRL 50X35 DATA-FILE PROGRAM
20 REM U.S.G.S. OPEN-FILE REPT. 79-1434
30 FIND #0,1
40 PRINT TAB10"INSTRUCTIONS FOR KRL 50X35 DATA-FILE PROGRAM"LIN3
50 PRINT TAB25"INTRODUCTION"LIN2
60 PRINT TAB6"This program permits the user to use a 50X35 matrix to store"
70 PRINT "split-precision (6-digit accuracy) data for up to 50 samples ('sets')"
80 PRINT "and up to 35 variables ('parameters') per sample, with user-defined"
90 PRINT "sample and parameter names. Once stored, the data may be edited,"
100 PRINT "printed out, added to, manipulated, and plotted via a number of"
110 PRINT "subroutines and plotting programs. If an HP-9871 impact printer is"
120 PRINT "available, the data may also be printed out in hard-copy table form,"
130 PRINT "including parameter names with superscript or subscript (i.e.,"
140 PRINT "isotope/oxide form) notation, with user-defined number of"
150 PRINT "significant figures, and with associated errors printed next to the"
160 PRINT "proper values."
170 PRINT LIN2,TAB22"GENERAL COMMENTS"LIN2
180 PRINT TAB6"If you intend to store a data-file, make sure a premarked,"
190 PRINT "empty or expendable file of at least 5000 words length is available."
200 PRINT
210 PRINT TAB6"Most displayed queries may be answered with a space in order to "
220 PRINT "escape from that subroutine."LIN1
230 PRINT TAB6"If an HP-9871 impact printer is connected to the computer and is"
240 PRINT "turned on, the data-printout will automatically be done on the"
250 PRINT "impact printer. The (standard) model 9866 thermal printer must have"
260 PRINT "a select code of 2, the impact printer a select code of 6 (assigned"
270 PRINT "in lines 50 and 60 of the program)."LIN3
280 PRINT TAB16"CREATING A DATA-FILE (OPTION 1)"LIN2
290 PRINT "DISPLAY: FILE NAME?"LIN1
300 PRINT "Input a name (<= 80 characters long) to be assigned to the file."LIN2
310 PRINT "DISPLAY: PARAMETER 1 NAME?"LIN1
320 PRINT TAB6"The parameter names may be up to 2 lines of 12 characters, so"
330 PRINT "this query requires two responses. A question-mark will prompt"
340 PRINT "input of the second line. If you wish the parameter name to be"
350 PRINT "printed out with all numerals as superscripts, input ^ as the first"
360 PRINT "character, first line (e.g., ^87Rb/86Sr). If numerals are to be"
370 PRINT "printed out as subscripts, input * as the first character of the"
380 PRINT "line (e.g., *A1203). Superscripts and subscripts will only appear"
390 PRINT "on impact-printer printouts and plotter-drawn plots. After each"
400 PRINT "parameter name has been entered, the display will request the name"
410 PRINT "of the next parameter. When no more parameters are to be named, "
420 PRINT "answer the two parameter-name queries with a space."LIN2
430 PRINT "DISPLAY: SET # 1 NAME?"LIN1
440 PRINT TAB6"The set (sample) names are stored as three lines of twelve char-"
450 PRINT "acters each, so that three display queries must be answered. When"
460 PRINT "no more sets are to be named, simply answer the first query with"
470 PRINT "a space. Set names are requested before the data for each set is"
480 PRINT "to be entered."LIN2
490 PRINT "DISPLAY: SET 1 , % A1203?"LIN1
500 PRINT TAB6"The parameter name here is of course an example. Input a value"
510 PRINT "for the particular set and parameter requested. The next query will"
520 PRINT "be the next-numbered parameter for that set, and so on until the"
```

530 PRINT "query will ask for data for the next-numbered set. To cease data-"
540 PRINT "input, enter a space in response to the first request for a set"
550 PRINT "name. If no value for this data-element is known or relevant, enter
560 PRINT "the number 1E50. The printout for this number will then be three"
570 PRINT "dashes (---)."LIN2
580 PRINT "DISPLAY: STORE IN TRACK#, FILE #?"LIN1
590 PRINT TAB6"If you do not wish to store the data yet, input a space."
600 PRINT "Otherwise, enter the track # and file # of the already-marked,"
610 PRINT ">5000-word file. Make sure the cassette tab is at the RECORD"
620 PRINT "position first, though."LIN2
630 PRINT TAB5"TO CREATE A DATA-FILE USING PARAMETERS DEFINED IN AN"
640 PRINT TAB15"EXISTING DATA-FILE (OPTION 0)"LIN2
650 PRINT TAB6"This option permits the use of parameters whose name and numbers
660 PRINT "have already been defined in the creation of a previous data-file."
670 PRINT "Operation is straightforward."LIN3
680 PRINT TAB12"PRINTOUT OF THE DATA-FILE (OPTIONS 4 AND 11)"LIN2
690 PRINT TAB6"Option 4 is a 'scratchpad' printout which prints out the data"
700 PRINT "as input, with numbered parameters and sets. Option 11 permits a"
710 PRINT "formal, hard-copy table to be produced."LIN1
720 PRINT "OPTION 4:"LIN1,-----"LIN1
730 PRINT "DISPLAY: DATA FROM TRACK#, FILE #?"LIN1
740 PRINT TAB6"If the data-file is already in memory, answer with a space."
750 PRINT "If not, make sure that the tape cartridge of interest is in the com-
760 PRINT "puter. Input the appropriate track and file numbers."LIN2
770 PRINT "DISPLAY: PARAMETER #?"LIN1
780 PRINT TAB6"You may print up to 4 parameters on a single sheet of paper"
790 PRINT "using the 9866 thermal printer, or up to 8 parameters using the 9871"
800 PRINT "impact printer and 15-inch wide paper. Enter the parameter num-"
810 PRINT "bers, separated by commas, that you wish printed out."LIN2
820 PRINT TAB6"If the data is being printed out on the impact printer, with"
830 PRINT "the number of significant figures defined by the user (option 11),"
840 PRINT "the uncertainties of each data-element can be printed out just to"
850 PRINT "the right of the data-element itself. The uncertainties must be"
860 PRINT "stored in the data-file as a specified parameter. In response"
870 PRINT "to the 'PARAMETER #' query, enter a slash (/) after each"
880 PRINT "appropriate parameter, followed by the parameter number of the "
890 PRINT "associated uncertainties. For example, an input of 2/17,6,3/19"
900 PRINT "would result in a printout of the uncertainties to parameters 2 and "
910 PRINT "3 (stored as parameters 17 and 19) being printed out, with a +/-"
920 PRINT "symbol, to the right of the appropriate data-elements, and with the"
930 PRINT "same number of decimal places as the data-element. If the uncer-"
940 PRINT "tainties are stored as percent, they will automatically be converted"
950 PRINT "to absolute values, so long as '%', 'PERCENT', or 'percent' appears"
960 PRINT "in the parameter name. Note that if too many characters are "
970 PRINT "required for the data-printout in this manner, interference with the"
980 PRINT "next column of data may occur. If the uncertainty is assigned a"
990 PRINT "value of zero, no uncertainty printout will occur."LIN2
1000 PRINT "DISPLAY: # OF SIGNIFICANT FIGURES?"LIN1
1010 PRINT TAB6"This display appears if you chose printout option 11 instead"
1020 PRINT "of 4. Enter the maximum number of significant figures (from 1 to 6)
1030 PRINT "that you wish printed out for each parameter chosen, in the order"
1040 PRINT "that you entered the parameter numbers. If all parameters are to"
1050 PRINT "have the same number of significant figures, you need only enter"
1060 PRINT "that number once. The data will be rounded off in the printout,"

1070 PRINT "with least significant digits of 5 rounded to the nearest even"
1080 PRINT "digit."
1090 PRINT TAB2"DISPLAY: SET(S) TO BE PRINTED OUT?"LIN1
1100 PRINT TAB6"If sets are to be printed out in serial order, input first"
1110 PRINT "last set numbers, separated by a comma. If sets are to be printed"
1120 PRINT "out in arbitrary order, input the set numbers in the desired order"
1130 PRINT "of printout, separated by semicolons. The two modes may be combined
1140 PRINT "in one response - e.g., an input of 2,5;24;11,9;16;1 would"
1150 PRINT "result in printout of sets in the order 2,3,4,5,24,11,10,9,16,1."
1160 PRINT "Note that sets may be printed out in descending as well as ascending
1170 PRINT "serial order. If 'ALL' or 'all' is the response to the query,"
1180 PRINT "all of the defined sets will be printed out in order. Data from"
1190 PRINT "different data-files (with the same parameters and parameter"
1200 PRINT "numbers) can be printed out in the same table by placing an asterisk
1210 PRINT "followed by the track and file numbers of the new data-file, before"
1220 PRINT "the set numbers of interest. Thus if sets 7 through 12 from the "
1230 PRINT "data-file in track 1, file 23 were to be added to the printout"
1240 PRINT "following sets 43 and 12 of a data-file in memory, the response to"
1250 PRINT "the 'SET(S)' query would be 43;12;*1,23,7,12."LIN2
1260 PRINT "OPTION 11: "LIN1"-----"LIN1
1270 PRINT "DISPLAY: PARAM. HEADINGS- 1=YES, 2=NO?"LIN3
1280 PRINT TAB6"An input of 1 will give a table with parameter and set names,"
1290 PRINT "but not parameter or set numbers, printed out. An input of 2 will"
1300 PRINT "suppress the printout of parameter names, in case data from more"
1310 PRINT "than one data-file is to be printed out, or if vertical juxta-"
1320 PRINT "position of different sheets of printout is desired."LIN2
1330 PRINT TAB20"TO EDIT DATA (OPTION 3)"LIN2
1340 PRINT "DISPLAY: EDIT SET #S?"LIN1
1350 PRINT TAB6"Enter the set numbers, separated by commas, of the sets to"
1360 PRINT "be edited. Thus if you wanted to change some of the values of"
1370 PRINT "sets 2, 41, and 16, the input would be 2,41,16."LIN2
1380 PRINT "DISPLAY: EDIT PARAMETER #S?"LIN1
1390 PRINT TAB6"Enter the parameter numbers whose values, for the sets chosen"
1400 PRINT "above, to be changed, separated by commas. To edit a sample name,"
1410 PRINT "0 must be the first number entered. To change all of the parameter"
1420 PRINT "values, enter ALL (or all). Thus to edit the sample name and all of"
1430 PRINT "the parameter values, the input would be 0,ALL. Enter a space "
1440 PRINT "to escape."
1450 PRINT LIN2
1460 PRINT TAB15"TO DELETE DATA SETS (OPTION 5)"LIN2
1470 PRINT "DISPLAY: SET(S) TO BE DELETED?"LIN1
1480 PRINT TAB6"If only one set is to be deleted, input that set number."
1490 PRINT "If a block of sets in sequence are to be deleted, input the first"
1500 PRINT "and last numbers of the sequence, with a comma separating the two."
1510 PRINT "The sets are renumbered after each deletion. Input a space after"
1520 PRINT "the desired deletions have been accomplished."LIN2
1530 PRINT TAB10"TO INSERT A DATA-SET (OPTION 6)"LIN2
1540 PRINT "DISPLAY: INSERT DATA BEFORE SET #?"LIN1
1550 PRINT TAB6"Input the number of the set before the new set is to be placed."
1560 PRINT "The usual data-input routine then occurs, and sets higher-numbered"
1570 PRINT "than the inserted set are renumbered."LIN2
1580 PRINT LIN1,TAB10"TO ADD A PARAMETER NAME (OPTION 9)"LIN1
1590 PRINT TAB6"This option permits you to increase the number of defined par-"
1600 PRINT "ameters by defining additional parameter names (up to 35 total)."

```
1610 PRINT "After adding a parameter name, all the values of the new parameter"  
1620 PRINT "will be zero. To input the corresponding data, you will need to:"  
1630 PRINT TAB10"CHANGE ALL VALUES OF ONE PARAMETER (OPTION 10)"LIN1  
1640 PRINT TAB6"This option permits you to change the values of a given param-"  
1650 PRINT "eter for all of the sets -- i.e., to enter data by column instead"  
1660 PRINT "of by row."LIN2  
1670 PRINT TAB5"TO PRINT OUT A LIST OF PARAMETER NAMES (OPTION 12)"LIN1  
1680 PRINT TAB5"TO PRINT OUT A LIST OF SET NAMES (OPTION 13)"LIN1  
1690 PRINT TAB5"TO PRINT OUT A LIST OF OPTIONS (OPTION 15)"LIN3  
1700 PRINT TAB6"These operations are straightforward."LIN3  
1710 PRINT LIN4  
1740 LINK #0,1  
1750 END
```

APPENDIX III -- LIST OF MAIN PROGRAM

```

10 COM AI[50,18],CI[35,13],DS[50,35],I2,M,B$(80)
20 REM DATA-FILE PROGRAM, K.R. LUDWIG, 8/79, USGS OPEN-FILE REPT. 79-1434
30 DIM GI[50],AS[36],CS[26],TS[72],ES[35],E$[36],SS[72],Z$[10],KI[9],HI[9]
40 DIM DS[150],FS[18],BI[9]
50 P1=2
60 P2=6
70 I0=50
80 M0=35
90 C=P1
100 STDPR P1
110 F$=" |-----"
120 SERROR A,4880
130 Z$="0123456789"
140 GOTO 400
150 DEF FNL(X)
160 E=1+POS(T$,",")
170 IF POS(Z$,T$[1,1]) THEN 200
180 T=-1
190 GOTO 210
200 T=VAL(T$)
210 IF POS(Z$,T$[E,E]) THEN 240
220 F=-1
230 GOTO 250
240 F=VAL(T$[E])
250 RETURN T
260 END
270 WRITE (P1,*)LIN1,SPA3,T$" -?? IMPROPER SYNTAX. TRY AGAIN. ***"LIN3
280 RETURN
290 WRITE (P1,*)"OBTAIN COMPLETE INSTRUCTIONS (-1)"
300 WRITE (P1,*)"CREATE USING 'OLD' PARAMETERS (0)"
310 WRITE (P1,*)"CREATE USING 'NEW' PARAMETERS (1)"
320 WRITE (P1,*)"ADD DATA TO EXISTING FILE (2)"LIN1"EDIT DATA (3)"
330 WRITE (P1,*)"PRINTOUT ('STANDARD' FORMAT) (4)"LIN1"DELETE DATA-SET (5)"
340 WRITE (P1,*)"INSERT DATA SET (6)"LIN1"EXCHANGE DATA-SETS (7)"
350 WRITE (P1,*)"EDIT PARAMETER NAME (8)"LIN1"ADD PARAMETER NAME (9)"
360 WRITE (P1,*)"CHANGE ALL VALUES OF ONE PARAMETER (10)"
370 WRITE (P1,*)"PRINTOUT IN FLEXIBLE FORMAT WITH RESTRICTED SIGN. FIGS. (11)"
380 WRITE (P1,*)"PRINTOUT LIST OF PARAMETER NAMES (12)"
390 WRITE (P1,*)"PRINTOUT LIST OF SAMPLE NAMES (13)"
400 WRITE (P1,*)"ENTER 15 TO OBTAIN LIST OF POSSIBLE OPERATIONS"LIN2
410 DISP "CREATE(1) ADD(2) EDIT(3) PRINT(4) DELETE(5) INSERT(6) EXCHANGE(7)";
420 INPUT F0
430 IF F0#-1 THEN 450
440 LINK #0,2
450 STANDARD
460 GOTO (F0=1)+2*(F0=15) OF 510,290
470 GOSUB 4200
480 IF F0=0 THEN 540
490 GOTO F0-1 OF 620,650,2490,1520,1720,1950,2260,2140,2200,2500,2360
500 GOTO F0=13+2*(F0#13) OF 2420,410
510 M=0
520 MAT D=ZER

```

```

530 MAT C=ZER
540 I2=0
550 MAT A=ZER
560 GOSUB 1210
570 IF NOT F0 THEN 590
580 GOSUB 1040
590 GOSUB 1240
600 GOSUB 980
610 GOTO 590
620 GOSUB 1240
630 GOSUB 980
640 GOTO 620
650 DISP "SET(S) TO BE EDITED";
660 INPUT SS
670 IF SS=" " THEN 410
680 SS[1+LEN(SS)]=",999"
690 TS=SS
700 IF FNLO>0 THEN 740
710 TS=TS[1,POS(TS,",999")-1]
720 GOSUB 270
730 GOTO 650
740 IF T=999 THEN 1420
750 I=T
760 IF I <= I2 THEN 790
770 GOSUB 4370
780 GOTO 650
790 SS=SS[1+POS(SS,",")]
800 DISP "SET" I "-EDIT PARAMETER #S";
810 INPUT TS
820 TS[1+LEN(TS)]=",999"
830 IF NOT POS(TS[1,3],"ALL") AND NOT POS(TS[1,3],"all") THEN 870
840 GOSUB 980
850 I2=I2-1
860 TS=TS[1+POS(TS,",")]
870 IF FNLO<0 OR T=999 THEN 690
880 IF T#0 THEN 910
890 GOSUB 1250
900 GOTO 960
910 J=T
920 IF J <= M THEN 950
930 GOSUB 2120
940 GOTO 800
950 GOSUB 1340
960 TS=TS[1+POS(TS,",")]
970 GOTO 830
980 FOR J=1 TO M
990 GOSUB 1340
1000 IF M=M0 THEN 410
1010 NEXT J
1020 I2=I2+1
1030 RETURN
1040 IF M <= M0 THEN 1070
1050 GOSUB 4350
1060 GOTO 410

```

```

1070 J=1+M
1080 GOSUB 1110
1090 GOTO 1070
1100 RETURN
1110 STANDARD
1120 DISP "PARAMETER "J"NAME";
1130 INPUT C$[1,13]
1140 INPUT C$[14,25]
1150 IF C$[14,19]#" THEN 1180
1160 IF C$[1,7]#" THEN 1180
1170 GOTO 1+(F0=1) OF 1200,590
1180 TRANSFER CS TO C[J,1]
1190 M=M+(F0#8)
1200 RETURN
1210 DISP "FILE NAME";
1220 INPUT B$
1230 RETURN
1240 I=I2+1
1250 IF I>IO THEN 1730
1260 DISP "SET#"I"NAME";
1270 INPUT A$[1,12]
1280 IF A$[1,12]="#" THEN 1420
1290 INPUT A$[13,24]
1300 IF A$[13,24]="#" THEN 1320
1310 INPUT A$[25,36]
1320 TRANSFER A$ TO A[I,1]
1330 RETURN
1340 TRANSFER C[J,1] TO CS
1350 DISP "SET#"I;C$[1+POS(CS[1,1],"^")+POS(CS[1,1],"*")];
1360 INPUT D[I,J]
1370 RETURN
1380 NEXT J
1390 I2=I2+1
1400 RETURN
1410 GOTO 1240
1420 DISP "STORE IN TRACK#, FILE#";
1430 INPUT TS
1440 B=2
1450 IF TS="#" THEN 410
1460 IF FNLO >= 0 AND F >= 0 THEN 1490
1470 GOSUB 270
1480 GOTO 1420
1490 STORE DATA #T,F
1500 B=0
1510 GOTO 410
1520 DISP "SET(S) TO BE DELETED";
1530 INPUT TS
1540 IF TS="#" THEN 1420
1550 IF FNLO>0 AND F>0 THEN 1580
1560 GOSUB 270
1570 GOTO 1520
1580 FOR K=1 TO 1+F-T
1590 GOSUB 1620
1600 NEXT K

```

```

1610 GOTO 1520
1620 FOR I=T TO I2
1630 IF I=I0 THEN 1700
1640 FOR J=1 TO M
1650 D[I,J]=D[I+1,J]
1660 NEXT J
1670 TRANSFER A[I+1,1] TO A$
1680 TRANSFER A$ TO A[I,1]
1690 NEXT I
1700 I2=I2-1
1710 RETURN
1720 IF I2<I0 THEN 1750
1730 PRINT TAB10"**** DATA-FILE IS FULL ****"LIN3
1740 GOTO 410
1750 DISP "INSERT DATA BEFORE SET #";
1760 INPUT TS
1770 IF TS=" " THEN 1420
1780 IF FNLO>0 AND F>0 THEN 1810
1790 GOSUB 270
1800 GOTO 1750
1810 FOR I=I2 TO T STEP -1
1820 FOR J=1 TO M
1830 D[I+1,J]=D[I,J]
1840 NEXT J
1850 TRANSFER A[I,1] TO A$
1860 TRANSFER A$ TO A[I+1,1]
1870 NEXT I
1880 I=T
1890 GOSUB 1250
1900 FOR J=1 TO M
1910 GOSUB 1340
1920 NEXT J
1930 I2=I+I2
1940 GOTO 1720
1950 DISP "EXCHANGE SET NUMBERS";
1960 INPUT TS
1970 IF TS=" " THEN 1420
1980 PRINT LIN3
1990 IF FNLO>0 AND (F>0 AND F <= I2) THEN 2020
2000 GOSUB 270
2010 GOTO 1950
2020 FOR J=1 TO M
2030 E[J]=D[T,J]
2040 D[T,J]=D[F,J]
2050 D[F,J]=E[J]
2060 NEXT J
2070 TRANSFER A[T,1] TO E$
2080 TRANSFER A[F,1] TO A$
2090 TRANSFER A$ TO A[T,1]
2100 TRANSFER E$ TO A[F,1]
2110 GOTO 1950
2120 WRITE (P1,*) TAB10"**** ONLY" M"PARAMETERS DEFINED ****"LIN3
2130 RETURN
2140 IF M<MO THEN 2170

```

```
2150 GOSUB 2120
2160 GOTO 410
2170 J=1+M
2180 GOSUB 1110
2190 GOTO 410
2200 DISP "PARAMETER #";
2210 INPUT J
2220 FOR I=1 TO I2
2230 GOSUB 1340
2240 NEXT I
2250 GOTO 400
2260 DISP "EDIT PARAMETER# ? NAME";
2270 INPUT TS
2280 IF TS=" " THEN 2350
2290 J=FNLO
2300 IF FNLO>0 AND F>0 THEN 2330
2310 GOSUB 270
2320 GOTO 2260
2330 J=T
2340 GOSUB 1110
2350 GOTO 410
2360 PRINT "PARAMETER #"TAB20"PARAMETER NAME"LIN2
2370 FOR J=1 TO M
2380 TRANSFER C[J,1] TO CS
2390 PRINT J;"-----";TAB18;CS,LIN3*(J=M)
2400 NEXT J
2410 GOTO 400
2420 PRINT "SAMPLE #"TAB20"SAMPLE NAME"LIN2
2430 FOR I=1 TO I2
2440 TRANSFER A[I,1] TO AS
2450 PRINT I;"-----";TAB17;AS,LIN3*(I=I2)
2460 NEXT I
2470 GOTO 400
2480 REM PRINTOUT ROUTINE
2490 K0=0
2500 C=P1*(STATP2<288)+P2*(STATP2 >= 288)
2510 STDPR C
2520 IF F0#4 THEN 4390
2530 Z=0
2540 DISP "PARAMETER NUMBERS";
2550 INPUT TS
2560 TS[1+LEN(TS)]=","
2570 FOR K=1 TO 4+(K0#0)+4*(C=P2)
2580 G[K]=FNLO
2590 IF G[K]<1 THEN 2780
2600 IF POS(TS[2,3],"/") AND C=P2 THEN 2630
2610 H[K]=0
2620 GOTO 2700
2630 L0=1+POS(TS,"/")
2640 IF POS(ZS[2],TS[L0,L0]) THEN 2670
2650 GOSUB 270
2660 GOTO 2530
2670 H[K]=VAL(TS[L0])
2680 TRANSFER C[H[K],1] TO CS
```

```

2690 B[K]= NOT (POS(C$, "%") + POS(C$, "percent") + POS(C$, "PERCENT"))
2700 IF H[K] <= M THEN 2720
2710 GOTO 2750
2720 T$=T$[1+POS(T$, ",")]
2730 IF G[K]<1 THEN 2820
2740 IF G[K] <= M THEN 2770
2750 GOSUB 2120
2760 GOTO 2530
2770 Z=Z+1
2780 NEXT K
2790 D$[1,150]=""
2800 E1=(H[1]#0)
2810 E3=1+4*((C=P2)*(Z<9)+(C=P1)*(Z<5))
2820 IF K0 THEN 2950
2830 PRINT "(parameter #)";
2840 STANDARD
2850 FOR K=1 TO Z
2860 IF C=P1 THEN 2890
2870 WRITE (C,2940)13,D$[1,10+E1+13*K];
2880 GOTO 2900
2890 PRINT TAB(9+13*K);
2900 PRINT G[K];
2910 NEXT K
2920 PRINT LIN2
2930 GOTO 3030
2940 FORMAT 10B
2950 DISP "# OF SIGNIFICANT FIGURES";
2960 INPUT T$
2970 FOR I=1 TO Z
2980 IF NOT POS(Z$[2],T$[1,1]) THEN 2950
2990 K[I]=VAL(T$)
3000 IF K[I]<1 OR K[I]>6 THEN 2950
3010 T$=T$[1+POS(T$, ",")]
3020 NEXT I
3030 IF K0=2 THEN 3350
3040 FOR S=0 TO 12 STEP 12
3050 IF C=P2 THEN 3070
3060 WRITE (C,2940)D$[1,(E3-1)* NOT S+14-S];
3070 FOR K=1 TO Z
3080 TRANSFER C[G[K],1] TO C$
3090 S0= NOT S*(1+POS(C$[1,1],"^") + POS(C$[1,1],"*")) + (S=12)*14
3100 IF C=P2 THEN 3130
3110 WRITE (C,2940)C$[S0,S0+11],32;
3120 GOTO 3190
3130 WRITE (C,2940)13,D$[1,13*K+E3],27,86,0,4;
3140 FOR L=S0 TO S0+11
3150 D=(POS(C$[1,1],"^") + 2*POS(C$[1,1],"*")) * (POS(Z$,C$[L,L])#0)
3160 WRITE (C,2940)27*(D=1),10*(D>0),C$[L,L],27*(D=2),10*(D>0);
3170 NEXT L
3180 WRITE (C,2940)27,86,0,16;
3190 NEXT K
3200 IF S=12 THEN 3260
3210 PRINT
3220 IF K0 THEN 3240

```

```

3230 PRINT "SET#";
3240 IF K0=2 THEN 3260
3250 PRINT TAB(F0=4)+E3"SAMPLE NAME";
3260 NEXT S
3270 PRINT
3280 IF K0 THEN 3300
3290 PRINT "----";
3300 FOR I=1 TO 1+Z
3310 WRITE (C,2940)FS[1+4*((I=1)*((F0=4)+(E3=1))+(I>1)),17+E1*(I=1)];
3320 NEXT I
3330 PRINT "|";LIN2
3340 GOTO 3350
3350 DISP "SET(S) TO BE PRINTED OUT";
3360 INPUT TS
3370 IF POS(T$,"all") OR POS(T$,"ALL") THEN 3530
3380 IF TS[1,1]#"*" THEN 3430
3390 TS=TS[2]
3400 GOSUB 4220
3410 K=1+POS(T$,"")
3420 TS=TS[K+POS(T$[K],",")]
3430 IF FNLO<0 THEN 4170
3440 TS[1+LEN(T$)]=";999"
3450 IF PCS(T$[2,3],",") THEN 3500
3460 IF T=999 THEN 4170
3470 F=FNLO
3480 TS=TS[1+POS(T$,";")]
3490 GOTO 3550
3500 GOTO (FNLO <= 0)+2*(F <= 0) OF 3350,3350,4170
3510 TS=TS[1+POS(T$,";")]
3520 GOTO 3550
3530 T=1
3540 F=I2
3550 FOR I=T TO F STEP (F >= T)-(F<T)
3560 IF I>I2 THEN 4170
3570 TRANSFER A[I,1] TO AS
3580 IF K0 THEN 3610
3590 STANDARD
3600 PRINT I;
3610 PRINT TAB1+E3-(K0#0);AS[1,12];
3620 FOR K=1 TO Z
3630 STANDARD
3640 N=D[I,G[K]]
3650 IF N#1E+50 THEN 3710
3660 IF C#P2 THEN 3690
3670 WRITE (C,2940)13,D$[1,K*13+E1+E3+3],"---";
3680 GOTO 4090
3690 PRINT TAB(13*K+E1+E3+3)"---";
3700 GOTO 4090
3710 IF K0 THEN 3750
3720 IF (ABSN >= 0.01)*(ABSN<1E+10)+ NOT N THEN 3880
3730 FLOAT 3
3740 GOTO 3880
3750 Q=K[K]
3760 IF N THEN 3790

```

```

3770 R=Q+1
3780 GOTO 3870
3790 IF ABSN >= 0.001 THEN 3840
3800 R=FNB(N/10^ (FNAN))
3810 R=Q+(R<Q)*(R-Q)
3820 GOSUB R OF 4780,4800,4820,4840
3830 GOTO 3880
3840 K7=10^(FNBN-FNAN-1)
3850 N=FNH(N*K7)/K7
3860 R=(Q-FNAN)*FNI(N/10)+ NOT FNI(N/10)
3870 GOSUB R OF 4620,4640,4660,4680,4700,4720,4740,4760
3880 IF C#P2 THEN 3930
3890 E2=K*13+E1+E3-1+2*( NOT H[K]*(F0=11) OR NOT K0*(N >= 0.01))
3900 DISP ""
3910 WRITE (C,2940)13,D$(1,E2);
3920 GOTO 3940
3930 PRINT TAB(13*K+E1+E3+2* NOT K0*( NOT N OR (ABSN >= 0.01)));
3940 IF H[K]=0 THEN 3970
3950 IF D[I,H[K]]=0 THEN 3970
3960 WRITE (C,2940)27,72,0,10;
3970 PRINT N;
3980 IF H[K]=0 THEN 4090
3990 IF D[I,H[K]]=0 THEN 4090
4000 WRITE (C,2940)27,86,0,3,27,10;
4010 Q9=R-(R=1)+(N<1)+FNA(10*N*(ABSN >= 1)+(ABSN<1))
4020 FOR J=1 TO 4+3*((Q9>3)+(Q9>6))-Q9
4030 WRITE (C,2940)8;
4040 NEXT J
4050 WRITE (C,2940)43,10,10,8,45,27,10;
4060 PRINT ABS(D[I,H[K]]*(B[K]+ NOT B[K]*D[I,G[K]]/100))
4070 WRITE (C,2940)27,10,8,8,8,27,86,0,16;
4080 WRITE (C,2940)27,72,0,12,27,10
4090 NEXT K
4100 IF A$(13,24)=" " THEN 4140
4110 PRINT LIN1,TAB1+E3-(K0#0);A$(13,24);
4120 IF A$(25,36)=" " THEN 4140
4130 PRINT LIN1,TAB1+E3-(K0#0);A$(25);
4140 PRINT LIN1
4150 NEXT I
4160 GOTO 3380
4170 PRINT LIN3
4180 STDPRT P1
4190 GOTO 410
4200 DISP "DATA FROM TRACK #, FILE #";
4210 INPUT TS
4220 B=1
4230 IF TS=" " THEN 4340
4240 IF FNLO#-1 AND F#-1 THEN 4270
4250 GOSUB 270
4260 GOTO 4200
4270 IF T#0 AND T#1 THEN 4250
4280 LOAD DATA #T,F
4290 B=0
4300 IF F0=11 AND C=P1 THEN 4340

```

```
4310 STANDARD
4320 WRITE (P1,*)"DATA FROM TRACK" T"FILE" F" ("I2"SETS,"M"PARAMETERS DEFINED)";
4330 WRITE (P1,*)LIN2;B$,LIN3
4340 RETURN
4350 WRITE (P1,*)TAB10"***** PARAMETER # LIMIT IS" M0"*****" LIN3
4360 RETURN
4370 WRITE (P1,*)TAB10"**** ONLY" I2"SETS DEFINED ****" LIN3
4380 RETURN
4390 DISP "PARAM. HEADINGS- (1)=YES, (2)=NO";
4400 INPUT K0
4410 GOTO 1+(K0=1)+2*(K0=2) OF 4390,4420,2530
4420 DISP "TITLE OF TABLE";
4430 INPUT D$
4440 PRINT LIN2,D$,LIN3
4450 GOTO 2530
4460 DEF FNA(N)=INTLGTABSN
4470 DEF FNB(N)
4480 S=1+FNAN
4490 N1=N
4500 N1=10*(ABSN1-AESINTN1)
4510 IF N1=0 THEN 4540
4520 S=S+1
4530 GOTO 4500
4540 RETURN S
4550 DEF FNI(N)=INTABSN#ABSN
4560 DEF FNH(N)
4570 R0=FNBN-Q
4580 K8=N/10^R0
4590 K9=K8-INTK8
4600 RETURN 10^R0*INT(K8+(K9>0.5)+(K9=0.5)*(INT(K8/2)#INTK8/2))
4610 END
4620 FIXED 0
4630 RETURN
4640 FIXED 1
4650 RETURN
4660 FIXED 2
4670 RETURN
4680 FIXED 3
4690 RETURN
4700 FIXED 4
4710 RETURN
4720 FIXED 5
4730 RETURN
4740 FIXED 6
4750 RETURN
4760 FIXED 7
4770 RETURN
4780 FLOAT 0
4790 RETURN
4800 FLOAT 1
4810 RETURN
4820 FLOAT 2
4830 RETURN
4840 FLOAT 3
```

```
4850 RETURN
4860 FLOAT 4
4870 RETURN
4880 GOTO A-55 OF 4910,4930,4930,4950,4990,5010,5030
4890 PRINT LIN2"***** ERROR"A"*****"LIN3
4900 GOTO 410
4910 PRINT LIN3"*** CANNOT FIND TRACK"FILE" F"-- TRY AGAIN ***"LIN3
4920 GOTO B OF 410,1420
4930 PRINT LIN3"*** TAPE IN PROTECT POSITION OR TAPE NOT IN SLOT ***"LIN3
4940 GOTO B OF 410,1420
4950 REWIND
4960 PRINT LIN3"*** TAPE VERIFICATION ERROR-- TRY AGAIN ***"LIN3
4970 GOTO B OF 410,1420
4980 F0=145+18*I0+13*M0+2*I0*M0
4990 PRINT LIN3"*** TRACK"FILE" F" IS TOO SMALL. MUST BE >"F0" WORDS. ***"LIN3
5000 GOTO 1420
5010 PRINT LIN3"*** TRACK"FILE" F" IS NOT A PROPER FILE FOR THIS PROGRAM **"LIN3
5020 GOTO B OF 410,1420
5030 PRINT LIN3"*** TRACK"FILE" F" IS NOT A DATA-FILE. TRY AGAIN ***"LIN3
5040 GOTO 410
5050 GOTO 4890
5060 END
```